Lab 1 Notes From Prof Calvin Plett

Getting Started

The SPICE deck listed on my web page should actually work as it is (previously it was lacking a probe command, but has now been added). However, there are some other things you will have to modify - but it is probably a good idea to verify it works first before spending a lot of time modifying it.

If the simulation fails with the error message "No Devices Found" This can occur if you have edited the file with Notepad or some other editor. You must make sure you edit the file directly from within PSPICE, then save the file as a .cir file. Do *NOT* edit with Notepad or other editor.

Modifications Needed




```
Q1 3 2 1 QN3904
*  C B E MODEL
VCC 4 0 DC 15V
* VOLTAGE SOURCE CALLED VCC CONNECTED FROM NODE 4 TO
* GROUND (NODE 0)
L1 4 3 159.0UH
* First Character specifies component
* ind, cap, res, bipolar tran, MOS tran, v src, I SRC, subcircuit, vcvs, vccs
*  L,    C,    R,         Q               M            V        I            X           E      G
*
* Next numbers specify nodes to which the component is connected
* Then component size or model name
* units: femto pico, micro, milli, kilo, meg, gig
*                F       P       U        M      K   MEG  G
* further letters behind units (eg V,A,H,F, OHMS) are ignored, but
* be careful, 3F is not 3 FARADS but 3 FEMTO FARADS,
* 1.5A is not 1.5 AMPERES but 1.5 ATTO (10^-18!!!).
C1 4 3 159PF
* No spaces between number and unit, else unit is ignored.
RL 4 3 3.00K
R1  2 0 5KOHMS
R2  4 2 10K
RE  1 0 5K
CE  1 0 100N
CIN 5 2 100N
* Matching components set close to 0
LMATCH 6 5 0.0001U
CMTCH1 5 0 0.0001P
CMTCH2 6 0 0.0001P
RIN 7 6 50
*V7 7 0 SIN(0 0.050 1.0MEG)
* SIN Source needed for transient analysis, here commented out
V7 7 0 DC 0 AC 1
* AC source needed for freq domain (freq resp, zin, zout, noise)
* This analysis is done on the equivalent linear model, so 1V is
```

Power Supply Voltage, needs to be changed to 5V

Adjust to set f_0

Should include RP (inductor parallel resistor)

Bias circuit needs adjusting for new VCC

Make sure coupling and bypass capacitors are correctly sized (large enough)

Matching, components initially close to zero

TRAN

AC



```
.PROBE
.OP
*
* Note changes made to correct model, add .PROBE, .OP (gives operating point info)
*
*.AC DEC 10 100K 10MEG
* AC analysis control line. Decade sweep 10 points per decade start at 100 kHz,
* end at 10 MHz, total is 2 decades, result is 21 points.
.AC LIN 41 600K 1.40MEG
* Linear sweep 600K to 1.4 MEG, total of 21 points. In 800k range, is 1 point/40k.
* result: no point at 900k
.NOISE V(3) V7 10
* Noise analyis V(3) is output noise summing node, V7 is input source
* every 10th point will cause a summary printout. Make sure this will give
* you a print out at your desired frequency
*.TRAN 20.N 5U
* point every 20 ns up to 5 u seconds.
* Transient sweep
* If your transient plot doesn't seem to be using the right steps,
* you can force it by using the "step ceiling value" as follows:
*.TRAN <print step value> <final time value> [no-print value
* [step ceiling value]][SKIPBP]
* Note - Need step ceiling value otherwise default step time too large
*.TRAN 10n 100u 0u 10n
.PRINT AC VDB(6) VP(6) VDB(3) VP(3)
.PRINT AC VM(6) VM(3) IR(V7) II(V7)
* VDB(6) voltage in dB on node 6, requires AC source somewhere
* VP is phase, VM magnitude, IM(V7) current magnitue through source V7
* VR real part of voltage, VI imaginary part,
.PRINT NOISE ONOISE INOISE
* ONOISE is output noise in volts per sqrt(Hz)
* INOISE is input referred noise in volts per sqrt(Hz)
.END
```


The design is built for power supply of 15V and center frequency of about 1 MHz. So, you will have to change the power supply voltage, and the components which determine the bias levels and the frequency of oscillation.

However, even without any changes, you should be able to run the SPICE file and get a result (just not the right result). Clearly if you change the frequency of oscillation, you will have to change the plotting frequency – the default is it plots from 600k to 1.4M (specified in the .AC command). Also, if you are going to zoom in more, or if the Q were to be increased, you might wish to see a smoother curve. To do this, you could use more points in the .AC command - it is now set at 41 points, there would be nothing wrong with 401 points or even way more to give you a smooth frequency response curve. In the transient response, it will sometimes plot fewer points than you ask for - you can force it to use all points by adding two more numbers behind your transient command as follows:

```
* Transient sweep
*.TRAN   [no-print value]
* [step ceiling value]][SKIPBP]
* Note - Need step ceiling value otherwise it may use the default
* step time which may be too large
.TRAN 10n 100u 0 10n
```

So, what do you need to do? Copy the spice file from the web page into your own file (you can use "file-->save page, then select directory, change the name to something appropriate with a .cir extension rather than .txt for example, filename.cir.

Running PSPICE, Selecting Traces to Display

To run, open PSPICE-AD (it might be part of EVAL 8, but this can change from year to year) then file --> open, find the directory, change the type to .cir, then load the file. You should be able to run the file directly by clicking on the little blue "run" arrow next to the file name.



The file should run almost instantaneously, and Probe should be opened automatically, showing an empty plot with frequency along the bottom. To see the results click on trace --> add trace --> v(3) --> OK and you will have a plot of output voltage as a function of frequency, showing a peak voltage of about 96V at about 980 kHz.



If you are printing this out, use a photo editor to invert the colours to avoid wasting ink, e.g.,



You can add extra plots, for example to have separate magnitude and phase plots as below. Note, cursors are shown and the little callout says peak gain at 980 kHz is 96 and the phase at 480 kHz is -177 degrees.



Note you can pick voltages, currents from the list, and you can add dB from the menu. Using dB, 96V becomes about 39.6 dB. Here I have marked the peak (using the cursor then plot → label → mark) then put the cursors on the -3dB points, at 36.6 and the bandwidth is shown as 325 kHz.



If you don't like your choice, from the plot window you can double click on the label at the bottom, e.g., V(3) and the Modify Trace window pops up and you can edit, or select another trace etc. Instead of selecting from the menu, you can type directly in the "Trace Expression" part of add or modify trace window. For example, vdb(3),vp(3) will result in node 3 voltage in dB or the phase of the voltage on node 3 in degrees.



if you are typing directly try:

```
dB        vdb()
phase     vp()
magnitude vm()
real      vr()
imaginary vi()
```

Displaying Equations, e.g., for Zin

You can also have PSPICE do more complex calculations. For example, to find input impedance, you need vin/iin and you can select this from the menu, or type in directly. For example vin/iin is V(6)/I(RIN) which you could ask for directly and ask for the real part, imaginary part, etc using equation functions

Some functions For equations:

```
magnitude M()
phase,    P()
real      R()
imaginary IMG()
```

Thus to get real and imaginary of the input impedance you can specify

```
R(V(6)/I(RIN)),IMG(V(6)/I(RIN)).
```



Note, the simulations shows you the impedance is 323.771 – j872.113. Obviously, impedance matching has not been done yet, or impedance would be 50 + j0. Note, for impedance matching, being within 10% or about 5Ω is close enough, that is 55 + j0, or 50 – j5, would be acceptable.

On occasion, you will need to select view → Output file which will open filename.out. This is useful in order to find errors, to see detailed operating point information (voltages, currents) or to do the noise analysis. This will be explained in more detail under noise analysis.

Design and Simulation steps:

(Also see marking scheme on web page)

DC Bias: Calculate resistors in Emitter, estimate input base current, choose bias current through R1 and R2 to be 10 times, knowing DC voltage on base and current through these resistors, calculate R1 and R2.

Bypass and coupling capacitors: knowing resistance you are bypassing or driving, choose C big enough so impedance is R/10 or preferably R/20. Bigger C is better, but in real life in the lab, a bigger C usually does not operate to as high frequency. Note that CE is bypassing re which is small, so CE will have to be quite large. CIN sees mainly R1 parallel with R2 parallel with rpi.

AC Components: Choose L in the required range, Find C to give the right frequency, find Rload to give the right bandwidth.

Simulate first to check DC levels, bias points. The easiest way is to check output file which lists DC voltages, currents through transistors. If DC is correct (and only if DC is correct) plot gain, find input impedance and noise figure. Note noise figure can be found from output listing (see comments on SPICE deck for more info.).

Then, match circuit, verify that matching is correct, then find gain, bandwidth, noise figure and compare to unmatched case.

Frequency Multiplication

Frequency Multiplier requires that you have harmonics, normally referred to as harmonic distortion. Normally, you would try to avoid distortion, but if you are designing the circuit to be frequency multiplier, then these harmonics are wanted. With a large input signal, the amplifier has enough distortion that it will work as it is (that is with no changes, except with a bigger input voltage) as a frequency multiplier. However, to enhance the nonlinearity, you could remove R2 and ground the emitter of the transistor. This will bias the base at 0 V, and only if a big input signal is applied (bigger than 0.7V) will current flow.

To see frequency multiplication, you need to run time domain simulations. This means you have to comment the .AC command and un-comment the .TRAN command. As well, you have to select the correct time-domain source (comment out the AC source, un-comment the sin source). And finally, you will need to change the input frequency such that it is a sub-multiple of the resonant frequency. The time domain simulation is shown on the next page.

To zoom in, change the x axis to 98-100us (plot → axis setting → user defined …)



We note, vin = 100mV p-p, vout = 7.0V p-p, gain = 70, gain is reduced due to nonlinearity.



Assuming the tuned circuit is resonant at 1 MHz, if a large input is applied at 500 kHz, then its second harmonic will happen at 1 MHz, and this frequency will be amplified by the tuned amplifier, while other harmonics and the fundamental at 500 kHz will be attenuated. Multiply by 2 is shown, with x axis changed to 96-100us to show 2 cycles of input.



Similarly, if the input is at 333 kHz, the third harmonic is at 1 MHz, so it will be amplified, and effectively, it is multiplying by 3 - Changed x axis to 92-100us to show 2 cycles of input.



If the Q is low, as above, the fundamental may still be quite large though, and the third harmonic is quite hard to see. To improve this, it is possible to increase the Q by increasing the load resistor, for example by 5 or 10 times. Below, RL has been changed by 5 times to 15k just to show the dramatic difference.



It would also be possible to do an FFT. More information is given below.

Examining the Output File and Noise Analysis

Noise analysis is done with ac analysis as described earlier. Then, the output is examined, with view → output file. Note that if you have modified the bias circuit or Q in order to see a better frequency multiplier, it should be brought back to its previous state before doing this analysis.

The output file shows the bias points, and the transistor operating point as below:

```
 ****     SMALL SIGNAL BIAS SOLUTION       TEMPERATURE =   27.000 DEG C

******************************************************************************

 NODE   VOLTAGE     NODE   VOLTAGE     NODE   VOLTAGE     NODE   VOLTAGE

(    1)    4.3172  (    2)    4.9873  (    3)   15.0000  (    4)   15.0000

(    5)    0.0000  (    6)    0.0000  (    7)    0.0000
```

```
**** BIPOLAR JUNCTION TRANSISTORS


NAME         Q1
MODEL        QN3904
IB           3.82E-06
IC           8.60E-04
VBE          6.70E-01
VBC         -1.00E+01
VCE          1.07E+01
BETADC       2.25E+02
GM           3.40E-02
RPI          6.85E+03
RX           6.69E+01
RO           1.54E+05
CBE          2.95E-11
CBC          1.31E-12
CJS          0.00E+00
BETAAC       2.33E+02
CBX/CBX2     6.26E-14
FT/FT2       1.75E+08
```

Scroll down until the noise data is found for your operating frequency, 1 MHz in this example:

```
****     NOISE ANALYSIS                TEMPERATURE =   27.000 DEG C
******************************************************************************
    FREQUENCY =  1.000E+06 HZ

 **** TRANSISTOR SQUARED NOISE VOLTAGES (SQ V/HZ)
        Q1
 RB       1.040E-14
 RC       7.329E-24
 RE       3.994E-17
 IBSN     1.557E-16
 IC       2.321E-15
 IBFN     0.000E+00
 TOTAL    1.291E-14

 **** RESISTOR SQUARED NOISE VOLTAGES (SQ V/HZ)
                     RL          R1           R2           RE          RIN
 TOTAL    4.739E-17  7.551E-17  3.775E-17  7.877E-20  7.543E-15

**** TOTAL OUTPUT NOISE VOLTAGE        =  2.062E-14 SQ V/HZ
                                       =  1.436E-07 V/RT HZ
     TRANSFER FUNCTION VALUE:

      V(3)/V7                  =  9.540E+01

     EQUIVALENT INPUT NOISE AT V7 =  1.505E-09 V/RT HZ
```

Noise factor is given by:

$$F = \frac{No,total}{No,source} = \frac{2.062 \times 10^{-14}}{7.543 \times 10^{-15}} = 2.66 \text{, or NF = 10 } \log_{10} 2.66 = 4.25 \text{ dB.}$$

Note SQ V/HZ means voltage squared per hertz. No,source is the noise due to resistor RIN, not the equivalent input noise at V7, that is the total noise at the input, not the noise due only to the source.

Other Potentially Useful PSPICE feature

- **To display a cross hair cursor** click the "toggle cursor" icon or use trace --> cursor --> display. Cursors can be moved by clicking at the desired location or by holding down the left button and dragging to a new location. Alternatively, the right and left arrows will move the cursor one position at a time, great for finding a peak exactly. A second cursor can be displayed by right clicking. This cursor can be moved with the arrows while holding down the shift key. The cursor is linked to a particular trace. To move to a different trace, left click or right click on the trace symbol at the bottom of the screen.
- **To mark a point on the plot** select cursor, find the point then click on the "Mark Label" icon, or Plot --> Label --> Mark. Labels can be erased by de-selecting cursors, clicking on the label and hitting delete.
- **To add a new plot** for example, to show magnitude and phase on separate plots. Use Plot --> Add Plot to Window. Then select the plot and use add trace and click on P and v(3) to get phase, or type in vp(3). An existing cursor can be moved to the new plot by left or right clicking on the diamond or square symbol preceding a trace name at the bottom of the new plot.
- **To save a particular customized plot style** use Window --> Display Control. Then type in an appropriate name, then Save. Next time you can load this style, (or click on Last Session), to avoid having to add all traces again, adjust them, etc.
- **To run an FFT** at any time, click on the fft icon. This mainly makes sense if doing a time domain simulation, to convert to frequency domain. Note, you should not include an initial transient, as typically you want steady state response. You can make sure you don't actually plot the first points by using a non-zero start time in your .TRAN command. Example below. The highest frequency you will see will be half of your sample frequency. For example, if your transient has time steps of 10nsec. that represents a sample frequency of 100 MHz, and your highest frequency will be 50 MHz. You may only be interested in a few MHz, in which case you would change your x axis setting to user defined and put in the desired maximum frequency. As well, if you run your simulation for a longer time you will get more resolution. For example, if your total simulation time (after intital transients) is 100 micro second, your resolution will be 1/100 usec = 10kHz. If you are having trouble seeing the harmonics change the Y axis to log (and you will have trouble unless you do this).

  Example Default command is: .TRAN 10n 100u 0 10n. This shows a start time of 0. You could examine the time domain output to see where to start, but with a narrow band amplifier, with a center at 1 MHz, it might take 20 usec, or even more to settle. As a futher check, you can run an FFT of your input sine wave. A pure sine wave should have a fundamental frequency, no harmonics, and a noise floor well below the fundamental.

To start the transient and hence the FFT at 20us, would change the .TRAN commant to .TRAN 10n 100u 20u 10n. A further observation is that a 1 MHz signal has a period of 1usec, thus with a step size of 10nsec, there are 100 points per period. Often more points work better, so could go to 1000 points per period to give a smoother output and a better looking FFT. To do this, the .TRAN command is changed .TRAN 1n 100u 20u 1n. Of course having step size of 1/10 results in a simulation that takes 10 times longer to run, but it should still just be a few seconds.

Possibly more to come later. Comments, questions welcomed.

email to: cp@doe.carleton.ca